INTERNATIONAL TELECOMMUNICATION UNION

ITU-T

TELECOMMUNICATION
STANDARDIZATION SECTOR
OF ITU

G.107
(12/98)

SERIES G: TRANSMISSION SYSTEMS AND MEDIA, DIGITAL SYSTEMS AND NETWORKS

International telephone connections and circuits – General definitions

---

The E-model, a computational model for use in transmission planning

ITU-T Recommendation G.107

(Previously CCITT Recommendation)

ITU-T G-SERIES RECOMMENDATIONS

TRANSMISSION SYSTEMS AND MEDIA, DIGITAL SYSTEMS AND NETWORKS

For further details, please refer to ITU-T List of Recommendations.

ITU-T RECOMMENDATION G.107

THE E-MODEL, A COMPUTATIONAL MODEL FOR USE IN TRANSMISSION PLANNING

Summary

This Recommendation gives the algorithm for the so-called E-model as the common ITU-T Transmission Rating Model. This computational model can be useful to transmission planners, to help ensure that users will be satisfied with end-to-end transmission performance. The primary output of the model is a scalar rating of transmission quality. A major feature of this model is the use of transmission impairment factors that reflect the effects of modern signal processing devices.

Source

ITU-T Recommendation G.107 was prepared by ITU-T Study Group 12 (1997-2000) and was approved under the WTSC Resolution No. 1 procedure on the 3rd of December 1998.

Keywords

Conversational quality, E-model, end-to-end transmission performance, impairment factors, transmission planning, transmission quality, transmission rating factor, transmission rating model, voice transmission quality mouth-to-ear.

FOREWORD

ITU (International Telecommunication Union) is the United Nations Specialized Agency in the field of telecommunications. The ITU Telecommunication Standardization Sector (ITU-T) is a permanent organ of the ITU. The ITU-T is responsible for studying technical, operating and tariff questions and issuing Recommendations on them with a view to standardizing telecommunications on a worldwide basis.

The World Telecommunication Standardization Conference (WTSC), which meets every four years, establishes the topics for study by the ITU-T Study Groups which, in their turn, produce Recommendations on these topics.

The approval of Recommendations by the Members of the ITU-T is covered by the procedure laid down in WTSC Resolution No. 1.

In some areas of information technology which fall within ITU-T's purview, the necessary standards are prepared on a collaborative basis with ISO and IEC.

NOTE

In this Recommendation the term *recognized operating agency (ROA)* includes any individual, company, corporation or governmental organization that operates a public correspondence service. The terms *Administration, ROA* and *public correspondence* are defined in the *Constitution of the ITU (Geneva, 1992).*

INTELLECTUAL PROPERTY RIGHTS

The ITU draws attention to the possibility that the practice or implementation of this Recommendation may involve the use of a claimed Intellectual Property Right. The ITU takes no position concerning the evidence, validity or applicability of claimed Intellectual Property Rights, whether asserted by ITU members or others outside of the Recommendation development process.

As of the date of approval of this Recommendation, the ITU had not received notice of intellectual property, protected by patents, which may be required to implement this Recommendation. However, implementors are cautioned that this may not represent the latest information and are therefore strongly urged to consult the TSB patent database.

CONTENTS

Recommendation G.107

THE E-MODEL, A COMPUTATIONAL MODEL FOR USE IN TRANSMISSION PLANNING

(Geneva, 1998)

1 General

1.1 Scope

This Recommendation describes a computational model, known as the E-model, that has proven useful as a transmission planning tool, for assessing the combined effects of variations in several transmission parameters that affect conversational[1] quality. This computational model can be used, for example, by transmission planners to help ensure that users will be satisfied with end-to-end transmission performance whilst avoiding over-engineering of networks. It must be emphasized that the primary output from the model is the "Rating Factor" R but this can be transformed to give estimates of customer opinion. Such estimates are only made for transmission planning purposes and not for actual customer opinion prediction (for which there is no agreed-upon model recommended by the ITU-T). Accordingly, the model described here is intended to be used to do relative comparisons of transmission conditions.

The E-model has not been fully verified by field surveys or laboratory tests for the very large number of possible combinations of input parameters. For many combinations of high importance to transmission planners, the E-model can be used with confidence, but for other parameter combinations, E-model predictions have been questioned and are currently under study. Accordingly, caution must be exercised when using the E-model for some conditions, for example, the E-model may give inaccurate results for combinations of certain types of impairments. Annex A provides further information in this regard.

1.2 References

The following ITU-T Recommendations and other references contain provisions which, through reference in this text, constitute provisions of this Recommendation. At the time of publication, the editions indicated were valid. All Recommendations and other references are subject to revision; all users of this Recommendation are therefore encouraged to investigate the possibility of applying the most recent edition of the Recommendations and other references listed below. A list of the currently valid ITU-T Recommendations is regularly published.

– ITU-T Recommendation G.113 (1996), *Transmission impairments.*

– CCITT Recommendation G.711 (1988), *Pulse Code Modulation (PCM) of voice frequencies.*

– ITU-T Recommendation G.723.1 (1996), *Speech coders: Dual rate speech coder for multimedia communications transmitting at 5.3 and 6.3 kbit/s.*

[1] Conversational quality in this context refers to transmission characteristics, e.g. long transmission times, effects of talker echoes, etc. However, the E-model as described in Recommendation is not intended to model transmission impairments during double talk situations.

– CCITT Recommendation G.726 (1990), *40, 32, 24, 16 kbit/s Adaptive Differential Pulse Code Modulation (ADPCM).*

– CCITT Recommendation G.727 (1990), *5- ,4-, 3- and 2-bits/sample embedded adaptive Differential Pulse Code Modulation (ADPCM).*

– CCITT Recommendation G.728 (1992), *Coding of speech at 16 kbit/s using low-delay code excited linear prediction.*

– ITU-T Recommendation G.729 (1996), *Coding of speech at 8 kbit/s using Conjugate-Structure Algebraic-Code-Excited Linear-Prediction (CS-ACELP).*

– ETSI ETS 300 580-2 (1994), *European digital cellular telecommunications system (Phase 2); Full rate speech; transcoding* (GSM 06.10).

– ETSI ETS 300 581-2 (1995), *European digital cellular telecommunications system; Half rate speech; Half rate speech transcoding* (GSM 06.20).

– ETSI ETS 300 726 (1996), *Digital cellular telecommunications system; Enhanced Full Rate (EFR) speech transcoding* (GSM 06.60).

– EIA/TIA/IS-54-B (1992*), Cellular System Dual-Mode Mobile Station-Base Station Compatibility Standard*.

2 The E-model, a computational model for use in transmission planning

2.1 Introduction

The complexity of modern networks requires that for transmission planning the many transmission parameters are not only considered individually but also that their combination effects are taken into account. This can be done by "expert, informed guessing," but a more systematic approach is desirable, such as by using a computational model. The output from the model described here is a scalar quality rating value, R, which varies directly with the overall conversational quality. Recommendation G.113 gives guidance about specific impairments, including combinations effects based upon a simplification of the model. However, the output can also give nominal estimates of user reactions, for instance in the form of percentages finding the modelled connection "Good or Better" or "Poor or Worse", as described in Annex B.

2.2 Source code

Annex C contains the source code in BASIC of the E-model described in this Recommendation. The purpose of this code is to ensure that users of the E-model are using consistent implementations of the formulae.

3 The structure and basic algorithms of the E-model

The E-model is based on the equipment impairment factor method, following previous transmission rating models. It was developed by an ETSI ad hoc group called "Voice Transmission Quality from Mouth to Ear".

The reference connection, as shown in Figure 1, is split into a send side and in a receive side. The model estimates the conversational quality from mouth to ear as perceived by the user at the receive side, both as listener and talker.



Figure 1/G.107 – Reference connection of the E-model

The transmission parameters used as an input to the computation model are shown in Figure 1. Values for room noise and for the *D*-factors are handled separately in the algorithm for send side and receive side and may be of different amount. The parameters SLR, RLR and circuit noise Nc are referred to a defined 0 dBr point. All other input parameters are either considered as values for the overall connection such as OLR (in any case the sum of SLR and RLR), number of qdu, equipment impairment factors *Ie* and expectation factor *A*, or referred only to the receive side, such as STMR, LSTR, WEPL (for calculation of Listener Echo) and TELR.

There are three different parameters associated with transmission time. The absolute delay *Ta* represents the total one-way delay between send side and receive side and is used to estimate the impairment due to too-long delay. The parameter mean one-way delay *T* represents the delay between the receive side (in talking state) and the point in a connection where a signal coupling occurs as a source of echo. The round-trip delay *Tr* only represents the delay in a 4-wire loop, where the "double reflected" signal will cause impairments due to Listener Echo.

3.1 Calculation of the transmission rating factor, *R*

According to the equipment impairment factor method, the fundamental principle of the E-model is based on a concept given in the description of the OPINE [11] model:

Psychological Factors on the psychological scale are additive.

The result of any calculation with the E-model in a first step is a transmission rating factor *R*, which combines all transmission parameters relevant for the considered connection. This rating factor *R* is composed of:

$$R = Ro - Is - Id - Ie + A \qquad (1)$$

Ro represents in principle the basic signal-to-noise ratio, including noise sources such as circuit noise and room noise. The factor *Is* is a combination of all impairments which occur more or less simultaneously with the voice signal. Factor *Id* represents the impairments caused by delay and the equipment impairment factor *Ie* represents impairments caused by low bit rate codecs. The expectation factor *A* allows for compensation of impairment factors when there are other advantages of access to the user. The term *Ro* and the *Is* and *Id* values are subdivided into further specific impairment values. The following subclauses give the formulae used in the E-model.

3.2 Basic signal-to-noise ratio, *Ro*

The basic signal-to-noise ratio *Ro* is defined by:

$$Ro = 15 - 1.5(SLR + No) \qquad (2)$$

The term *No* [in dBm0p] is the power addition of different noise sources:

$$No = 10\,\lg\left[10^{\frac{Nc}{10}} + 10^{\frac{Nos}{10}} + 10^{\frac{Nor}{10}} + 10^{\frac{Nfo}{10}}\right] \qquad (3)$$

Nc [in dBm0p] is the sum of all circuit noise powers, all referred to the 0 dBr point.

Nos [in dBm0p] is the equivalent circuit noise at the 0 dBr point, caused by the room noise *Ps* at the send side:

$$Nos = Ps - SLR - Ds - 100 + 0.008(Ps - OLR - Ds - 14)^2 \qquad (4)$$

where $OLR = SLR + RLR$. In the same way the room noise *Pr* at the receive side is transferred into an equivalent circuit noise *Nor* [in dBm0p] at the 0 dBr point.

$$Nor = RLR - 121 + Pre + 0.008(Pre - 35)^2 \qquad (5)$$

The term *Pre* [in dBm0p] is the "effective room noise" caused by the enhancement of *Pr* by the listener's sidetone path:

$$Pre = Pr + 10\,\lg\left[1 + 10^{\frac{(10 - LSTR)}{10}}\right] \qquad (6)$$

Nfo [in dBm0p] represents the "noise floor" at the receive side,

$$Nfo = Nfor + RLR \qquad (7)$$

with *Nfor* usually set to –64 dBmp.

3.3 Simultaneous impairment factor, *Is*

The factor *Is* is the sum of all impairments which may occur more or less simultaneously with the voice transmission. The factor *Is* is divided into three further specific impairment factors:

$$Is = Iolr + Ist + Iq \qquad (8)$$

Iolr represents the decrease in quality caused by too-low values of OLR and is given by:

$$Iolr = 20\left[\left\{1+\left(\frac{X}{8}\right)^{8}\right\}^{\frac{1}{8}} - \frac{X}{8}\right] \tag{9}$$

where:

$$X = OLR + 0.2(64 + No - RLR) \tag{10}$$

The factor *Ist* represents the impairment caused by non-optimum sidetone:

$$Ist = 10\left[1+\left(\frac{STMRo-12}{5}\right)^{6}\right]^{\frac{1}{6}} - 46\left[1+\left(\frac{STMRo}{23}\right)^{10}\right]^{\frac{1}{10}} + 36 \tag{11}$$

where:

$$STMRo = -10\,\lg\left[10^{-\frac{STMR}{10}} + \mathrm{e}^{-\frac{T}{4}}10^{-\frac{TELR}{10}}\right] \tag{12}$$

The impairment factor *Iq* represents impairment caused by quantizing distortion:

$$Iq = 15\,\lg\left[1+10^{Y}\right] \tag{13}$$

where:

$$Y = \frac{R_o - 100}{15} + \frac{46-G}{10} \tag{14}$$

and:

$$G = 1.07 + 0.258Q + 0.0602Q^2 \tag{15}$$

$$Q = 37 - 15\,\lg(\mathrm{qdu}) \tag{16}$$

In this formula qdu means the number of qdu for the whole connection between send side and receive side.

NOTE – If an impairment factor *Ie* is used for a piece of equipment, then the qdu value for that same piece of equipment must not be used.

3.4 Delay impairment factor, *Id*

Also *Id*, the impairment factor representing all impairments due to delay of voice signals is further subdivided into the three factors *Idte*, *Idle* and *Idd*:

$$Id = Idte + Idle + Idd \tag{17}$$

The factor *Idte* gives an estimate for the impairments due to Talker Echo:

$$Idte = \left[\frac{Roe-Re}{2} + \sqrt{\frac{(Roe-Re)^2}{4} + 100} - 1\right]\left(1-\mathrm{e}^{-T}\right) \tag{18}$$

where:

$$Roe = -1.5(No - RLR) \tag{19}$$

$$Re = 80 + 2.5(TERV - 14) \quad (20)$$

$$TERV = TELR - 40 \lg \frac{1 + \frac{T}{10}}{1 + \frac{T}{150}} + 6e^{-0.3T^2} \quad (21)$$

For values of $T < 1$ ms, the Talker Echo should be considered as sidetone, i.e. $Idte = 0$. The computation algorithm furthermore combines the influence of STMR to Talker Echo. Taking into account that low values of STMR may have some masking effects on the Talker Echo and for very high values of STMR the Talker Echo may become more noticeable, the terms *TERV* and *Idte* are adjusted as follows:

For STMR < 9 dB:

In equation (20) *TERV* is replaced by *TERVs*, where:

$$TERVs = TERV + \frac{Ist}{2} \quad (22)$$

For 9 dB ≤ STMR ≤ 15 dB:

the above given equations (18) to (21) apply.

For STMR > 15 dB:

Idte is replaced by *Idtes*, where:

$$Idtes = \sqrt{Idte^2 + Ist^2} \quad (23)$$

The factor *Idle* represents impairments due to Listener Echo. The equations are:

$$Idle = \frac{Ro - Rle}{2} + \sqrt{\frac{(Ro - Rle)^2}{4} + 169} \quad (24)$$

where:

$$Rle = 10.5(WEPL + 7)(Tr + 1)^{-0.25} \quad (25)$$

The factor *Idd* represents the impairment caused by too-long absolute delay *Ta*, which occurs even with perfect echo cancelling.

For $Ta < 100$ ms:

$$Idd = 0$$

For $Ta > 100$ ms:

$$Idd = 25\left\{\left(1 + X^6\right)^{\frac{1}{6}} - 3\left(1 + \left[\frac{X}{3}\right]^6\right)^{\frac{1}{6}} + 2\right\} \quad (26)$$

with:

$$X = \frac{\lg\left(\frac{Ta}{100}\right)}{\lg 2} \quad (27)$$

3.5 Equipment impairment factor, *Ie*

The values for the Equipment Impairment Factor *Ie* of elements using low bit rate codecs are not related to other input parameters. They are depending on subjective mean opinion score test results as well as on network experience. Some values listed in Table 1 are taken from Table 7/G.113. Refer to Recommendation G.113 to determine if more recent information is available.

Table 1/G.107 – Planning values for the equipment impairment factor Ie

Codec type	Operating rate kbit/s	Value Ie	Reference
ADPCM	40	2	G.726, G.727
	32	7	G.721 (1988), G.726, G.727
	24	25	G.726, G.727
	16	50	G.726, G.727
LD-CELP	16	7	G.728
	12.8	20	
CS-ACELP	8	12 (Note)	G.729
	8	13 (Note)	G.729-A + VAD
VSELP	8	20	IS-54-B, TIA
RPE-LTP	13	20	GSM 06.10, Full-rate
VSELP	5.6	23	GSM 06.20, Half-rate
ACELP	12.2	3 (Note)	GSM 06.60, Enhanced full-rate
ACELP	5.3	19 (Note)	G.723.1
MP-MLQ	6.3	15 (Note)	G.723.1
CELP+	6.8	25	
NOTE – Provisionally.			

3.6 Expectation factor, *A*

Due to the specific meaning of the expectation factor *A*, there is – consequently – no relation to all other transmission parameters. Some provisional values are given in Table 2.

Table 2/G.107 – Provisional examples for the expectation factor A

Communication system example	Maximum value of A
Conventional (wirebound)	0
Mobility by cellular networks in a building	5
Mobility in a geographical area or moving in a vehicle	10
Access to hard-to-reach locations, e.g. via multi-hop satellite connections	20

It should be noted that the values in Table 2, taken from Recommendation G.113, are only provisional. The use of the factor *A* and its selected value in a specific application is up to the planner's decision. However, the values in Table 2 should be considered as absolute upper limits for *A*.

3.7 Default values

For all input parameters used in the algorithm of the E-model, the default values are listed in Table 3. It is strongly recommended to use these default values for all parameters which are not varied during planning calculation. If all parameters are set to the default values, the calculation results in a very high quality with a rating factor of $R = 94.2$.

Table 3/G.107 – Default values and permitted ranges for the parameters

Parameter	Abbr.	Unit	Default value	Permitted range	Remark
Sending Loudness Rating	SLR	dB	+8	0 ... +18	Note 1
Receiving Loudness Rating	RLR	dB	+2	–5 ... +14	Note 1
Sidetone Masking Rating	STMR	dB	15	10 ... 20	Note 2
Listener Sidetone Rating	LSTR	dB	18	13 ... 23	Note 2
D-Value of Telephone, Send Side	Ds	–	3	–3 ... +3	Note 2
D-Value of Telephone Receive Side	Dr	–	3	–3 ... +3	Note 2
Talker Echo Loudness Rating	TELR	dB	65	5 ... 65	
Weighted Echo Path Loss	WEPL	dB	110	5 ... 110	
Mean one-way Delay of the Echo Path	T	msec	0	0 ... 500	
Round Trip Delay in a 4-wire Loop	Tr	msec	0	0 ... 1000	
Absolute Delay in echo-free Connections	Ta	msec	0	0 ... 500	
Number of Quantization Distortion Units	qdu	–	1	1 ... 14	
Equipment Impairment Factor	I_e	–	0	0 ... 40	
Circuit Noise referred to 0 dBr-point	Nc	dBm0p	–70	–80 ... –40	
Noise Floor at the Receive Side	Nfor	dBmp	–64	–	Note 3
Room Noise at the Send Side	Ps	dB(A)	35	35 ... 85	
Room Noise at the Receive Side	Pr	dB(A)	35	35 ... 85	
Expectation Factor	A	–	0	0 ... 20	

NOTE 1 – Total values between microphone or receiver and 0 dBr-point.

NOTE 2 – Fixed relation: LSTR = STMR + D.

NOTE 3 – Currently under study.

ANNEX A

Examples of conditions where caution must be exercised when using the E-model

NOTE – When the results of Question 20/12 (1997-2000) on the assessment of the E-model are available, they will be included here. These results are expected to be available in late 1999 or early 2000.

1) *The overall level of the equipment impairment factors*

Some experimental investigations suggest that the general tendency of the equipment impairment factors is too pessimistic, so that a hidden security margin may be incorporated.

2) *The overall additivity property of the model*

The E-model supposes that different kinds of impairments are additive on the scale of the transmission rating factor *R*. This feature has not been checked to a satisfying extent. Especially, very few investigations are available regarding the interaction of low bit rate codecs with other kinds of impairments, e.g. with room noise. Additionally, the order effects when tandeming several low bit rate codecs remain uncertain.

3) *The coverage of talker sidetone*

Some experiments show that the E-model disregards some masking effects occurring for talker sidetone, namely in conjunction with circuit noise, room noise at receive side, and low delay talker echo (<10 ms).

4) *The effect of room noise at send side*

Several investigations show that the E-model disregards the Lombard effect, i.e. the fact that the speaker adopts his/her pronunciation and speaking level to the noise environment. This leads to too pessimistic E-model predictions for high room noise levels *Pr*.

5) *The expectation factor A*

Up to now it has not been clarified under which conditions the given values for the expectation factor should be applied. It is expected that these values may depend, e.g. on the user group, and that the absolute values will change in long term.

6) *Predictions for quantizing distortion and the derivation methodology for new equipment impairment factors*

This feature is not very important for transmission planning, but for the derivation of new equipment impairment factors. Subjective test results for MNRU reference conditions are very often more pessimistic than E-model predictions. Thus, the transformation according to the "equivalent Q method" described in Recommendation G.113 may put the test results in the saturation, and thereby influence the derived impairment factors.

7) *Predictions for different types of room noise, and different frequency shapes in the communication channel, in the sidetone path and in the echo path*

The E-model regards the effect of room noise only by means of an A-weighted level. The actual opinion on the speech communication quality may depend even on the type and disturbance of the environmental noise. The frequency characteristics of the communication channel, of the sidetone and of the echo path are not explicitly regarded by the E-model, but only implicitly by means of loudness ratings. However, they may affect the perceived transmission quality.

ANNEX B

Quality measures derived from the transmission rating factor *R*

The transmission rating factor *R* can lie in the range from 0 to 100, where $R = 0$ represents an extremely bad quality and $R = 100$ represents a very high quality. The E-model provides a statistical estimation of quality measures. The percentages for a judgement Good or Better (GOB) or Poor or Worse (POW) are obtained from the *R*-factor by means of the Gaussian Error function:

$$E(x) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{x} e^{-\frac{t^2}{2}} dt \tag{B.1}$$

The equations are:

$$GOB = 100E\left(\frac{R-60}{16}\right) \quad \% \tag{B.2}$$

$$POW = 100E\left(\frac{45-R}{16}\right) \quad \% \tag{B.3}$$

The Mean Opinion Score (MOS) in the scale 1-5 can be obtained from the *R*-factor using the formulae:

For $R < 0$: $\quad \mathrm{MOS} = 1$

For $0 < R < 100$: $\quad \mathrm{MOS} = 1 + 0.035R + R(R-60)(100-R)\,7 \cdot 10^{-6}$ (B.4)

For $R > 100$: $\quad \mathrm{MOS} = 4.5$

GOB, POW and MOS as functions of *R* are depicted in Figures B.1 and B.2 respectively.



Figure B.1/G.107 – GOB (Good or Better) and POW (Poor or Worse) as functions of rating factor *R*



Figure B.2/G.107 – MOS as function of rating factor *R*

In some cases, transmission planners may not be familiar with the use of quality measures such as the *R* rating factor obtained from planning calculations, and thus provisional guidance for interpreting calculated *R* factors for planning purposes is given in Table B.1[2]. This table also contains equivalent transformed values of *R* into MOS, GOB and POW.

Table B.1/G.107 – Provisional guide for the relation between *R*-value and user satisfaction

***R*-value (lower limit)**	**MOS (lower limit)**	**GOB (%) (lower limit)**	**POW (%) (upper limit)**	**User satisfaction**
90	4.34	97	~0	Very satisfied
80	4.03	89	~0	Satisfied
70	3.60	73	6	Some users dissatisfied
60	3.10	50	17	Many users dissatisfied
50	2.58	27	38	Nearly all users dissatisfied

[2] The source of Table B.1 is Table 3/G.175.

ANNEX C

Source code

```
1  CLS
2  PRINT "PROGRAM g107_1"
3  REM THIS VERSION IS CONFORM WITH THE ALGORITHM
4  REM DESCRIBED IN REC. G.107
5  REM PROGRAM WRITTEN BY N.O. JOHANNESSON
6  REM LAST MODIFICATIONS BY S. MOELLER, APRIL 1998
7  PRINT
8  PRINT "E-model, algorithm according to ITU-T Rec. G.107 Annex C,"
9  PRINT "for voice communication between side (S) and (R)."
10 PRINT
11 PRINT "The E-model gives an estimation of voice transmission quality"
12 PRINT "mouth-to-ear as perceived at Side (R). The model output is a"
13 PRINT "transmission rating factor R which can be transformed to give"
14 PRINT "estimates of customer opinion. Such estimates are only made for"
15 PRINT "transmission planning purposes and not for actual customer"
16 PRINT "opinion prediction. See ITU-T Rec. G.107 for more details."
17 PRINT
18 PRINT "Note: To continue program:"
19 PRINT "a) INPUT appropriate value as indicated on screen,"
20 PRINT "or"
21 PRINT "b) if no value is asked for, PRESS THE RETURN KEY."
22 PRINT
23 DIM X(100), Y(5, 100), Col(5), Pxo(5)
24 REM Set parameters at default values
25 GOSUB 29
26 REM Goto main menu
27 GOTO 199


29 REM Default Parameter Values, according to Tab.3/G.107 (lines 29-50)
30 SLR = 8
31 RLR = 2
32 STMRs = 15
33 Ds = 3
34 STMR = 15
35 Dr = 3
36 LSTRr = 18
37 TELR = 65
38 T = 0
39 WEPL = 110
40 Tr = 0
41 Ta = 0
42 Ie = 0
43 A = 0
44 Nc = -70
45 Ps = 35
46 Pr = 35
47 qdu = 1
48 Nfor = -64
50 RETURN


198 REM Main menu (lines 198-370)
199 PRINT
200 PRINT "Print current parameter values ", "=1"
210 PRINT "Input new parameter values     ", "=2"
```

```
220 PRINT "Compute table, one parameter   ", "=3"
230 PRINT "Set parameter at default values", "=4"
240 PRINT "Exit program                   ", "=5"
250 PRINT
260 INPUT Y1
270 CLS
280 IF Y1 = 1 THEN GOSUB 500
290 IF Y1 = 2 THEN GOSUB 1000
300 IF Y1 = 3 THEN GOSUB 2000
310 IF Y1 = 4 THEN GOSUB 30
320 IF Y1 = 5 THEN GOTO 9999
330 CLS
340 IF Y1 = 4 THEN PRINT , "Parameters set at default values !"
350 GOTO 199


500 REM SUB Print current parameter values (lines 500-700)
510 PRINT , "SLR="; SLR, "RLR="; RLR, "OLR= SLR + RLR="; SLR + RLR
520 PRINT , "Side (S): Ds="; Ds
530 PRINT , "Side (R): STMR="; STMR, "Dr="; Dr, "LSTR="; STMR + Dr
540 PRINT
550 PRINT , "TELR="; TELR, "Mean One-way Delay T ms="; T
560 PRINT , "WEPL="; WEPL, "Round-trip Delay Tr ms="; Tr
570 PRINT , "One-way Absolute Delay Ta ms="; Ta
580 PRINT
590 PRINT , "Noise Floor at Side (R) Nfor dBmp="; Nfor
600 PRINT , "Circuit Noise Nc dBm0p="; Nc
610 PRINT , "Room Noise, Side (S), Ps dB(A)="; Ps
620 PRINT , "Room Noise, Side (R), Pr dB(A)="; Pr
630 PRINT
640 PRINT , "qdu="; qdu
650 PRINT
660 PRINT , "Equipment Impairment Factor Ie="; Ie
670 PRINT , "Expectation Factor A="; A
680 PRINT
690 INPUT C$
700 RETURN


1000 REM SUB Input Parameters (lines 1000-1270)
1020 CLS
1030 PRINT "Type designation of parameter for which the value is to be changed !"
1031 PRINT
1032 PRINT "Note 1. New value of OLR is obtained indirectly, i.e. by new"
1033 PRINT "value of SLR or RLR. (OLR=SLR+RLR.)"
1034 PRINT
1035 PRINT "Note 2. New value of LSTR is obtained indirectly, i.e. by new"
1036 PRINT "value of STMR or Dr. (LSTR=STMR+Dr.)"
1037 PRINT
1040 INPUT "Parameter:"; A$
1050 INPUT "New Value="; Px
1060 PRINT A$; "="; Px
1070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px
1080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px
1090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN STMR = Px
1100 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN Dr = Px
1110 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px
1120 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px
1130 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px
1140 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px
1150 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px
```

```
1160 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px
1170 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px
1180 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px
1190 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px
1200 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px
1210 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px
1220 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px
1230 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px
1240 PRINT
1250 IF Y1 = 2 THEN INPUT "More parameters changed, Yes(1) or No(0)"; Ypar
1260 IF Ypar = 1 THEN GOTO 1020
1270 RETURN


2000 REM SUB Tabulate (lines 2000-3000)
2020 INPUT "Variable Parameter:"; A$
2030 PRINT "(To exit tabulation, put parameter value = 1000 !)"
2040 PRINT TAB(8); A$; TAB(18); "R"; TAB(28); "GOB %"; TAB(38); "POW %"; TAB(48);
"MOS"
2050 INPUT Px
2060 IF Px = 1000 THEN GOTO 3000
2070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px
2080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px
2090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN
2100   STMR = Px
2110   LSTR = STMR + Dr
2120 END IF
2130 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN
2140   Dr = Px
2150   LSTR = STMR + Dr
2160 END IF
2170 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px
2180 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px
2190 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px
2200 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px
2210 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px
2220 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px
2230 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px
2240 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px
2245 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px
2250 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px
2260 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px
2270 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px
2280 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px
2290 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px
2300 GOSUB 3500
2400 GOSUB 4000
2500 GOSUB 4100
2600 GOSUB 4200
2700 R = INT(R * 10 + .5) / 10
2800 PRINT TAB(8); Px; TAB(18); R; TAB(28); GOB; TAB(38); POW; TAB(48); MOS
2900 GOTO 2050
3000 RETURN

3500 REM Compute R (lines 3500-3880)

3509 REM Noise Summation, formulas (3) to (7)
3510 Nr1 = Ps - SLR - Ds - 100
3520 Nr1 = Nr1 + .008 * (Ps - SLR - RLR - Ds - 14) ^ 2
3530 LSTR = STMR + Dr
3540 Pro = Pr + 10 * LOG(1 + 10 ^ ((10 - LSTR) / 10)) / LOG(10)
```

```
3550 Pr1 = Pro + .008 * (Pro - 35) ^ 2
3560 Nr2 = Pr1 - 121 + RLR
3570 Nfo = Nfor + RLR
3580 No = 10 * LOG(10 ^ (Nr1 / 10) + 10 ^ (Nr2 / 10) + 10 ^ (Nc / 10) + 10 ^ (Nfo
/ 10)) / LOG(10)
3590 Nt = No - RLR

3599 REM Ro, formula (2)
3600 Ro = 15 - 1.5 * (SLR + No)

3609 REM Iolr, formulas (9) and (10)
3610 Xolr = SLR + RLR + .2 * (64 + Nt)
3620 Iolr = 20 * ((1 + (Xolr / 8) ^ 8) ^ (1 / 8) - Xolr / 8)

3629 REM Ist, formulas (11) and (12)
3630 STMRo = -10 * LOG(10 ^ (-STMR / 10) + 10 ^ (-TELR / 10) * EXP(-T / 4)) /
LOG(10)
3640 Ist = 10 * (1 + ((STMRo - 12) / 5) ^ 6) ^ (1 / 6) - 10
3650 Ist = Ist - 46 * (1 + (STMRo / 23) ^ 10) ^ (1 / 10) + 46

3659 REM Iq, formulas (13) to (16)
3660 IF qdu < 1 THEN qdu = 1
3670 Q = 37 - 15 * LOG(qdu) / LOG(10)
3680 G = 1.07 + .258 * Q + .0602 * Q ^ 2
3690 Iq = 15 * LOG(1 + 10 ^ ((Ro - 100) / 15) * 10 ^ ((46 - G) / 10)) / LOG(10)

3699 REM Is, formula (8)
3700 Isyn = Iolr + Ist + Iq

3709 REM TERV, formula (21)
3710 TERV = TELR + 6 * EXP(-.3 * T ^ 2) - 40 * LOG((1 + T / 10) / (1 + T / 150))
/ LOG(10)

3719 REM Modifications to satisfy formula (22)
3720 IF STMR < 9 THEN TERV = TERV + .5 * Ist

3729 REM Idte, formulas (18) to (20)
3730 Re = 80 + 2.5 * (TERV - 14)
3740 Roe = -1.5 * (No - RLR)
3750 Xdt = (Roe - Re) / 2
3760 Idte = Xdt + SQR(Xdt ^ 2 + 100)
3770 Idte = (Idte - 1) * (1 - EXP(-T))

3779 REM Modifications to satisfy formula (23)
3780 IF STMR > 15 THEN Idte = SQR(Idte ^ 2 + Ist ^ 2)

3789 REM Idle, formulas (24) and (25)
3790 Rle = 10.5 * (WEPL + 7) * (Tr + 1) ^ (-1 / 4)
3800 Xdl = (Ro - Rle) / 2
3810 Idle = Xdl + SQR(Xdl ^ 2 + 169)

3819 REM Idd, formulas (26) and (27)
3820 IF Ta < 100 THEN Idd = 0
3830 IF Ta = 100 THEN Idd = 0
3840 IF Ta > 100 THEN
        X = (LOG(Ta / 100)) / LOG(2)
        Idd = 25 * ((1 + X ^ 6) ^ (1 / 6) - 3 * (1 + (X / 3) ^ 6) ^ (1 / 6) + 2)
3850 END IF

3859 REM Id
3860 Id = Idte + Idle + Idd
```

```
3869 REM R, formula (1)
3870 R = Ro - Isyn - Id - Ie + A
3880 RETURN

4000 REM Compute GOB, formula (B.2) (lines 4000-4050)
4010 Z# = (R - 60) / 16
4020 GOSUB 5000
4030 GOB = 100 * F#
4040 GOB = INT(GOB * 10 + .5) / 10
4050 RETURN

4100 REM Compute POW, formula (B.3) (lines 4100-4150)
4110 Z# = (R - 45) / 16
4120 GOSUB 5000
4130 POW = 100 * (1 - F#)
4140 POW = INT(POW * 10 + .5) / 10
4150 RETURN

4200 REM Compute MOS, formula (B.4) (lines 4200-4260)
4210 MOS = 1 + R * .035 + R * (R - 60) * (100 - R) * 7 * 10 ^ (-6)
4220 MOS = INT(MOS * 100 + .5) / 100
4230 IF R < 0 THEN MOS = 1
4240 IF MOS < 1 THEN MOS = 1
4250 IF R > 100 THEN MOS = 4.5
4260 RETURN

5000 REM Norm Distr F(Z), formula (B.1) (lines 5000-5130)
5010 S# = 0
5020 N% = 0
5030 H# = Z#
5040 S# = S# + H#
5050 H# = H# * (-1) * (Z#) ^ 2 * (2 * N% + 1) / ((N% + 1) * 2 * (2 * N% + 3))
5060 N% = N% + 1
5070 IF ABS(H#) < 10 ^ (-6) THEN GOTO 5090
5080 GOTO 5040
5090 S# = S# / (SQR(2 * 3.14159265#))
5100 F# = .5 + S#
5110 F# = INT(F# * 10 ^ 5 + .5) / 10 ^ 5
5120 REM PRINT "Z="; Z#, "F(Z)="; F#, "N="; N%
5130 RETURN

9999 END
```

APPENDIX I

Bibliography

– Supplement 3 to ITU-T Series P Recommendations (1993), *Models for predicting transmission quality from objective measurements.*

ITU-T RECOMMENDATIONS SERIES

Series A	Organization of the work of the ITU-T
Series B	Means of expression: definitions, symbols, classification
Series C	General telecommunication statistics
Series D	General tariff principles
Series E	Overall network operation, telephone service, service operation and human factors
Series F	Non-telephone telecommunication services
Series G	**Transmission systems and media, digital systems and networks**
Series H	Audiovisual and multimedia systems
Series I	Integrated services digital network
Series J	Transmission of television, sound programme and other multimedia signals
Series K	Protection against interference
Series L	Construction, installation and protection of cables and other elements of outside plant
Series M	TMN and network maintenance: international transmission systems, telephone circuits, telegraphy, facsimile and leased circuits
Series N	Maintenance: international sound programme and television transmission circuits
Series O	Specifications of measuring equipment
Series P	Telephone transmission quality, telephone installations, local line networks
Series Q	Switching and signalling
Series R	Telegraph transmission
Series S	Telegraph services terminal equipment
Series T	Terminals for telematic services
Series U	Telegraph switching
Series V	Data communication over the telephone network
Series X	Data networks and open system communications
Series Y	Global information infrastructure
Series Z	Programming languages